

#T23

Пакетные запросы Временные таблицы. Работа с запросами в коде.

Программист 1С: с нуля до профи




Из этой лекции вы узнаете

- Что такое **пакетный запрос**
- Для чего в запросах используются **временные таблицы**
- Как работать с запросами в **программном коде**

Пакетные запросы

Обычный запрос



Пакетный запрос

Сервер 1С

№ пакета

0

```
ВЫБРАТЬ *  
ИЗ Справочник.Номенклатура КАК T  
;
```

1

```
ВЫБРАТЬ *  
ИЗ Справочник.Контрагенты КАК T  
;
```

2

```
ВЫБРАТЬ *  
ИЗ Справочник.Организации КАК T
```

Обработка массива
результатов запроса

Сервер СУБД

Выполнить запрос и
вернуть результат

№ пакета	Результат запроса
0	РезультатПакета0
1	РезультатПакета1
2	РезультатПакета2


Обычный запрос

Что надо купить:



Дом

Магазин




Пакетный запрос

Что надо купить:



Дом



Магазин



Пакетный запрос. Пример

ВЫБРАТЬ

Номенклатура.Код КАК Код,
Номенклатура.Наименование КАК Наименование,
Номенклатура.Артикул КАК Артикул,
Номенклатура.СтавкаНДС КАК СтавкаНДС

ИЗ

Справочник.Номенклатура КАК Номенклатура

;


////////////////////////////////////
ВЫБРАТЬ

Контрагенты.Код КАК Код,
Контрагенты.Наименование КАК Наименование,
Контрагенты.ИНН КАК ИНН,
Контрагенты.ВидКонтрагента КАК ВидКонтрагента,
Контрагенты.Покупатель КАК Покупатель,
Контрагенты.Поставщик КАК Поставщик,
Контрагенты.ИсточникПривлечения КАК ИсточникПривлечения

ИЗ

Справочник.Контрагенты КАК Контрагенты


Пакетный запрос. Конструктор запроса



Временные таблицы

Временные таблицы

Результат выполнения запроса можно «сохранить»,
поместив его во **временную** таблицу:



Временные таблицы

- **Предназначение:**
 - Временное хранение данных при выполнении запроса
- Позволяют сделать текст сложных запросов более легким для восприятия
- Создаются по умолчанию в оперативной памяти
 - СУБД может при необходимости помещать временную таблицу в служебную БД «TempDB»

Временные таблицы. Секция **ПОМЕСТИТЬ**

ВЫБРАТЬ

Номенклатура.Код КАК Код,
Номенклатура.Наименование КАК Наименование,
Номенклатура.Артикул КАК Артикул

ПОМЕСТИТЬ ТаблицаТоваров

ИЗ

Справочник.Номенклатура КАК Номенклатура

Временные таблицы. Пример

Вложенный
запрос



```
ВЫБРАТЬ
    Контакты.Значение КАК Значение
ИЗ
    (ВЫБРАТЬ
        КонтрагентыКонтактнаяИнформация.Значение КАК Значение
    ИЗ
        Справочник.Контрагенты.КонтактнаяИнформация КАК КонтрагентыКонтактнаяИнформация
    ГДЕ
        КонтрагентыКонтактнаяИнформация.Тип = &Тип
    ОБЪЕДИНИТЬ ВСЕ
    ВЫБРАТЬ
        КонтактныеЛицаКонтактнаяИнформация.Значение
    ИЗ
        Справочник.КонтактныеЛица.КонтактнаяИнформация КАК КонтактныеЛицаКонтактнаяИнформация
    ГДЕ
        КонтактныеЛицаКонтактнаяИнформация.Тип = &Тип) КАК Контакты

СГРУППИРОВАТЬ ПО
    Контакты.Значение
```

Временная
таблица



```
ВЫБРАТЬ
    КонтрагентыКонтактнаяИнформация.Значение КАК Значение
ПОМЕСТИТЬ ВТ_Контакты
ИЗ
    Справочник.Контрагенты.КонтактнаяИнформация КАК КонтрагентыКонтактнаяИнформация
ГДЕ
    КонтрагентыКонтактнаяИнформация.Тип = &Тип

ОБЪЕДИНИТЬ ВСЕ

ВЫБРАТЬ
    КонтактныеЛицаКонтактнаяИнформация.Значение
ИЗ
    Справочник.КонтактныеЛица.КонтактнаяИнформация КАК КонтактныеЛицаКонтактнаяИнформация
ГДЕ
    КонтактныеЛицаКонтактнаяИнформация.Тип = &Тип
;

////////////////////////////////////
ВЫБРАТЬ
    Контакты.Значение КАК Значение
ИЗ
    ВТ_Контакты КАК Контакты

СГРУППИРОВАТЬ ПО
    Контакты.Значение
```

Временные таблицы. Пример

```
ВЫБРАТЬ
    Телефоны.Значение КАК Значение
ПОМЕСТИТЬ ВТ_Контакты
ИЗ
    Справочник.Контрагенты.КонтактнаяИнформация КАК Телефоны
ГДЕ
    Телефоны.Тип = &Тип
ОБЪЕДИНИТЬ ВСЕ
ВЫБРАТЬ
    Телефоны.Значение
ИЗ
    Справочник.КонтактныеЛица.КонтактнаяИнформация КАК Телефоны
ГДЕ
    Телефоны.Тип = &Тип
;
////////////////////////////////////
ВЫБРАТЬ
    Контакты.Значение КАК Значение
ИЗ
    ВТ_Контакты КАК Контакты
СГРУППИРОВАТЬ ПО
    Контакты.Значение
```

Конструктор запроса

Таблицы и поля | Группировка | Условия | Дополнительно | Объединения/Псевдонимы | Индекс | Построитель | Пакет запросов

Выборка записей

Первые

Без повторяющихся

Разрешенные

Тип запроса

Выборка данных

Создание временной таблицы

Уничтожение временной таблицы

Имя временной таблицы:

ВТ_Контакты


Блокировать получаемые данные для последующего изменения

Таблицы

Контрагенты

Таблицы для изменения

Временные таблицы. Пример



Важная мысль

Крайне нежелательно помещать во временную таблицу очень много строк

- Помещение большого количества строк во временную таблицу может привести долгому выполнению запроса, нагрузке на оборудование (память, диски)

Пример задачи

Информация об остатках товаров хранится в регистре накопления «Остатки товаров». Информация о ценах товаров хранится в регистре сведений «Цены номенклатуры». Нужно получить таблицу с информацией об **остатках** и **ценах всех** товаров. Склад «Основной», вид цен «Розничная».

Наименование	Артикул	Остаток	Цена
Монитор	M11	10	150
Клавиатура	K120	100	11

Работа с запросами в коде

Конструкторы

Только текст запроса

```
"ВЫБРАТЬ
| Номенклатура.Ссылка КАК Ссылка
|ИЗ
| Справочник.Номенклатура КАК Номенклатура"
```

Текст запроса + код для обработки результата

```
///}{КОНСТРУКТОР_ЗАПРОСА_С_ОБРАБОТКОЙ_РЕЗУЛЬТАТА
// Данный фрагмент построен конструктором.
// При повторном использовании конструктора, внесенные вручную

Запрос = Новый Запрос;
Запрос.Текст =
    "ВЫБРАТЬ
    | Номенклатура.Ссылка КАК Ссылка
    |ИЗ
    | Справочник.Номенклатура КАК Номенклатура";

РезультатЗапроса = Запрос.Выполнить ();

ВыборкаДетальныеЗаписи = РезультатЗапроса.Выбрать ();


Пока ВыборкаДетальныеЗаписи.Следующий () Цикл
    // Вставить обработку выборки ВыборкаДетальныеЗаписи
КонецЦикла;

///}}КОНСТРУКТОР_ЗАПРОСА_С_ОБРАБОТКОЙ_РЕЗУЛЬТАТА
```

&НаСервере

Функция ВыполнитьЗапрос ()

Конец



Общая схема выполнения запроса



1. Установка текста запроса



Запрос (Query)

Текст (Text)

Использование:

Чтение и запись.

Описание:

Тип: [Строка](#).

Содержит исходный текст выполняемого запроса.

Доступность:

Сервер, толстый клиент, внешнее соединение, мобильное приложение (сервер), мобильный автономный сервер.

```
Запрос = Новый Запрос;
```

```
Запрос.Текст =
```

```
"ВЫБРАТЬ
```

```
| Номенклатура.Ссылка КАК Ссылка,
```

```
| Номенклатура.Наименование КАК Наименование,
```

```
| Номенклатура.Артикул КАК Артикул
```

```
| ИЗ
```

```
| Справочник.Номенклатура КАК Номенклатура
```

```
| ГДЕ
```

```
| Номенклатура.Ссылка В ИЕРАРХИИ (&ГруппаСправочника) " ;
```

2. Установка параметров запроса

Параметры
запроса

2

Запрос

Запрос (Query)

Параметры (Parameters)

Использование:

Только чтение.

Описание:

Тип: [Структура](#).

Содержит значения параметров, установленных запросу.

Доступность:

Сервер, толстый клиент, внешнее соединение, мобильное приложение (сервер), мобильный автономный сервер.

```
Запрос = Новый Запрос;
```

```
Запрос.Текст =
```

```
"ВЫБРАТЬ
```

```
| Номенклатура.Ссылка КАК Ссылка,
```

```
| Номенклатура.Наименование КАК Наименование,
```

```
| Номенклатура.Артикул КАК Артикул
```

```
| ИЗ
```

```
| Справочник.Номенклатура КАК Номенклатура
```

```
| ГДЕ
```


```
| Номенклатура.Ссылка В ИЕРАРХИИ (&ГруппаСправочника) " ;
```

```
Группа = Справочники.Номенклатура.НайтиПоНаименованию ("Телевизоры") ;
```

```
Запрос.УстановитьПараметр ("ГруппаСправочника", Группа) ; ←
```



3. Выполнение запроса



РезультатЗапроса = Запрос.Выполнить ();

4. Получение выборки из результата запроса

```
Выборка = РезультатЗапроса.Выбрать ( <ТипОбхода> );  
//обход записей в выборке  
Пока Выборка.Следующий () Цикл  
    //обращение к полям  
    Сообщить (Выборка.Наименование) ;  
КонецЦикла;
```

ОбходРезультатаЗапроса (QueryResultIteration)

Значения

[ПоГруппировкам \(ByGroups\)](#)

[ПоГруппировкамСИерархией \(ByGroupsWithHierarchy\)](#)

[Прямой \(Linear\)](#)

Описание:


Содержит варианты обхода результата запроса.

Доступность:

Сервер, толстый клиент, внешнее соединение.
Возможен обмен с сервером.

Типы обхода выборки:

- **Прямой** (по умолчанию)
- **ПоГруппировкам**
- **ПоГруппировкамСИерархией**



5.1. Выгрузка результата запроса в таблицу значений

Как выгрузить результат запроса в **таблицу значений**:

```
Запрос = Новый Запрос;  
Запрос.Текст =  
"ВЫБРАТЬ  
| Номенклатура.Ссылка КАК Ссылка,  
| Номенклатура.Наименование КАК Наименование,  
| Номенклатура.Артикул КАК Артикул  
|ИЗ  
| Справочник.Номенклатура КАК Номенклатура  
|ГДЕ  
| Номенклатура.Ссылка В ИЕРАРХИИ(&ГруппаСправочника)";  
  
Группа = Справочники.Номенклатура.НайтиПоНаименованию("Телевизоры");  
  
Запрос.УстановитьПараметр("ГруппаСправочника", Группа);  
  
РезультатЗапроса = Запрос.Выполнить();  
  
ТаблицаДанных = РезультатЗапроса.Выгрузить();
```



5.2. Выгрузка результата запроса в дерево значений

Как выгрузить результат запроса в **дерево значений**:

```
Запрос = Новый Запрос;  
Запрос.Текст =  
"ВЫБРАТЬ  
|   Номенклатура.Ссылка КАК Ссылка,  
|   Номенклатура.Наименование КАК Наименование,  
|   Номенклатура.Артикул КАК Артикул  
|ИЗ  
|   Справочник.Номенклатура КАК Номенклатура  
|ГДЕ  
|   Номенклатура.Ссылка В ИЕРАРХИИ (&ГруппаСправочника)  
|ИТОГИ ПО  
|   Номенклатура.Родитель";  
  
Группа = Справочники.Номенклатура.НайтиПоНаименованию ("Телевизоры");  
  
Запрос.УстановитьПараметр ("ГруппаСправочника", Группа);  
  
РезультатЗапроса = Запрос.Выполнить ();  
  
ДеревоДанных = РезультатЗапроса.Выгрузить (ОбходРезультатаЗапроса.ПоГруппировкам);
```



Линейный (прямой) порядок обхода

```
Запрос = Новый Запрос;  
Запрос.Текст =  
    "ВЫБРАТЬ  
    |   Взаиморасчеты.Контрагент КАК Контрагент,  
    |   Взаиморасчеты.Договор КАК Договор,  
    |   Взаиморасчеты.СуммаОстаток КАК Долг  
    |ИЗ  
    |   РегистрНакопления.ВзаиморасчетыСКонтрагентами.Остатки КАК Взаиморасчеты  
    |ИТОГИ  
    |   СУММА (СуммаОстаток)  
    |ПО  
    |   Контрагент ИЕРАРХИЯ";  
  
РезультатЗапроса = Запрос.Выполнить ();  
  
Выборка = РезультатЗапроса.Выбрать ();  
  
Пока Выборка.Следующий () Цикл  
    Сообщить ("Контрагент: " + Выборка.Контрагент  
        + " Договор: " + Выборка.Договор  
        + " Долг: " + Выборка.Долг);  
  
КонецЦикла;
```

— Контрагент: ЛорДорМех Договор: Долг: -9 760	ИТОГИ	
— Контрагент: ЛорДорМех Договор: № 1 от 01.02.2010 Долг: -2 720		детали
— Контрагент: ЛорДорМех Договор: № 2 от 05.06.2011 Долг: -7 040		
— Контрагент: Алютех ООО Договор: Долг: -9 285	ИТОГИ	
— Контрагент: Алютех ООО Договор: Основной от 16.04.2020 Долг: -9 285		детали

Обход по группировкам (без детальных записей)

```
Запрос = Новый Запрос;  
Запрос.Текст =  
    "ВЫБРАТЬ  
    |   Взаиморасчеты.Контрагент КАК Контрагент,  
    |   Взаиморасчеты.Договор КАК Договор,  
    |   Взаиморасчеты.СуммаОстаток КАК Долг  
    |ИЗ  
    |   РегистрНакопления.ВзаиморасчетыСКонтрагентами.Остатки КАК Взаиморасчеты  
    |ИТОГИ  
    |   СУММА (СуммаОстаток)  
    |ПО  
    |   Контрагент ИЕРАРХИЯ";  
  
РезультатЗапроса = Запрос.Выполнить ();  
  
ВыборкаКонтрагент = РезультатЗапроса.Выбрать (ОбходРезультатаЗапроса.ПоГруппировкам);  
  
Пока ВыборкаКонтрагент.Следующий() Цикл  
  
    Сообщить ("Контрагент: " + ВыборкаКонтрагент.Контрагент  
        + " Договор: " + ВыборкаКонтрагент.Договор  
        + " Долг: " + ВыборкаКонтрагент.Долг);  
  
КонецЦикла;
```

— Контрагент: ЛорДорМех Договор: Долг: -9 760

— Контрагент: Алютех ООО Договор: Долг: -9 285

Обход по группировкам (с детальными записями)

```

Запрос = Новый Запрос;
Запрос.Текст =
"ВЫБРАТЬ
|   Взаиморасчеты.Контрагент КАК Контрагент,
|   Взаиморасчеты.Договор КАК Договор,
|   Взаиморасчеты.СуммаОстаток КАК Долг
|ИЗ
|   РегистрНакопления.ВзаиморасчетыСКонтрагентами.Остатки КАК Взаиморасчеты
|ИТОГИ
|   СУММА (СуммаОстаток)
|ПО
|   Контрагент ИЕРАРХИЯ";

РезультатЗапроса = Запрос.Выполнить ();

ВыборкаКонтрагент = РезультатЗапроса.Выбрать (ОбходРезультатаЗапроса.ПоГруппировкам);

Пока ВыборкаКонтрагент.Следующий () Цикл

    Сообщить ("Контрагент: " + ВыборкаКонтрагент.Контрагент
        + " Долг: " + ВыборкаКонтрагент.Долг);

    ВыборкаДетальныеЗаписи = ВыборкаКонтрагент.Выбрать ();
    Пока ВыборкаДетальныеЗаписи.Следующий () Цикл
        Сообщить ("---Контрагент: " + ВыборкаДетальныеЗаписи.Контрагент
            + " Договор: " + ВыборкаДетальныеЗаписи.Договор
            + " Долг: " + ВыборкаДетальныеЗаписи.Долг);
    КонецЦикла;

КонецЦикла;

```

— Контрагент: ЛорДорМех Долг: -9 760	ИТОГИ	
— ---Контрагент: ЛорДорМех Договор: № 1 от 01.02.2010 Долг: -2 720		детали
— ---Контрагент: ЛорДорМех Договор: № 2 от 05.06.2011 Долг: -7 040		
— Контрагент: Алютех ООО Долг: -9 285	ИТОГИ	
— ---Контрагент: Алютех ООО Договор: Основной от 16.04.2020 Долг: -9 285		детали

Обход по группировкам с иерархией

```
«НаСервере
□ Процедура СформироватьНаСервере ()

    Запрос = Новый Запрос;
    Запрос.Текст =
        "ВЫБРАТЬ
        |   Взаиморасчеты.Контрагент КАК Контрагент,
        |   Взаиморасчеты.Договор КАК Договор,
        |   Взаиморасчеты.СуммаОстаток КАК Долг
        |ИЗ
        |   РегистрНакопления.ВзаиморасчетыСКонтрагентами.Остатки КАК Взаиморасчеты
        |ИТОГИ
        |   СУММА (СуммаОстаток)
        |ПО
        |   Контрагент ИЕРАРХИЯ";

    РезультатЗапроса = Запрос.Выполнить ();

    Выборка = РезультатЗапроса.Выбрать (ОбходРезультатаЗапроса.ПоГруппировкамСИерархий);

    ВыбратьВсеВложения (Выборка);

КонецПроцедуры

«НаСервере
□ Процедура ВыбратьВсеВложения (ИерархическаяВыборка)

    СпособВыборки = ОбходРезультатаЗапроса.ПоГруппировкамСИерархий;
    Пока ИерархическаяВыборка.Следующий () Цикл

        Сообщить ("Контрагент: " + ИерархическаяВыборка.Контрагент
            + " Договор: " + ИерархическаяВыборка.Договор
            + " Долг: " + ИерархическаяВыборка.Долг);

        Если ИерархическаяВыборка.ТипЗаписи() = ТипЗаписиЗапроса.ИтогПоИерархии Тогда
            ДочерняяВыборка = ИерархическаяВыборка.Выбрать (СпособВыборки, ИерархическаяВыборка.Группировка ());
        Иначе
            ДочерняяВыборка = ИерархическаяВыборка.Выбрать (СпособВыборки);
        КонецЕсли;

        ВыбратьВсеВложения (ДочерняяВыборка);

    КонецЦикла;

КонецПроцедуры // ()
```

Обход по группировкам с иерархией

— Контрагент: ЛорДорМех Договор: № 1 от 01.02.2010 Долг: -2 720		детали
— Контрагент: ЛорДорМех Договор: № 2 от 05.06.2011 Долг: -7 040		
— Контрагент: Покупатели Договор: Долг: -9 285		итог по иерархии
— Контрагент: Алютех ООО Договор: Долг: -9 285		итог по группировке
— Контрагент: Алютех ООО Договор: Основной от 16.04.2020 Долг: -9 285		детали

Конец 23-го занятия

Подписывайтесь на нас в социальных сетях 😊



<https://www.youtube.com/channel/UC124tAuP11GlaGpK3YGHzw>



<https://www.facebook.com/ironskillsby/>



https://www.instagram.com/ironskills_/